
Abstract
We present alternative approaches to routing and scheduling in Answer Set Programming (ASP), and

explore them in the context of Multi-agent Path Finding. The idea is to capture the flow of time in

terms of partial orders rather than time steps attached to actions and fluents. This also abolishes the

need for fixed upper bounds on the length of plans. The trade-off for this avoidance is that (parts of)

temporal trajectories must be acyclic, since multiple occurrences of the same action or fluent cannot be

distinguished anymore. While this approach provides an interesting alternative for modeling routing,

it is without alternative for scheduling since fine-grained timings cannot be represented in ASP in a

feasible way. This is different for partial orders that can be efficiently handled by external means such as

acyclicity and difference constraints. We formally elaborate upon this idea and present several resulting

ASP encodings. Finally, we demonstrate their effectiveness via an empirical analysis.

1. Introduction

The ease of Answer Set Programming (ASP [1]) to express reachability has made it a prime

candidate for addressing various routing problems, like multi-agent path finding [2], phylo-

genetic inference [3], wire routing [4], etc. This lightness vanishes, however, once routing is

combined with scheduling for expressing deadlines and durations since fine-grained timings

cannot be feasibly represented in ASP. This is because ASP [5], just like CP [6] and SAT [7],

usually account for time by indexing action and fluent variables with time steps, a technique

tracing back to situation calculus [8] and temporal logic [9]. Each time step results in a copy of

the problem description. Hence, the finer the granularity of time, the more copies are produced.

Although this is usually a linear increase, it eventually leads to a decrease in performance.

We rather capture flows of time by means of partial orders on actions and/or fluents, similar

to partial-order planning [10]. In fact, the avoidance of time steps also eliminates the need for

upper bounds on temporal trajectories, that is, horizons or makespans. The trade-off for this is

that (parts of) temporal trajectories must be acyclic, since multiple occurrences of the same

action or fluent cannot be distinguished without indexing. Moreover, to cease the influence of

the granularity of time on the solving process, the idea is to outsource the treatment of partial

orders by using hybrid ASP, more precisely, acyclicity and difference constraints. Intuitively,

these constraints are used for ordering actions in view of collision evasion. As a matter of fact,

we have already applied this technique in several industrial-scale applications involving routing

and scheduling, namely, train scheduling [11], system design [12], and warehouse robotics [13].

However, the intricacy of these applications obscured a clear view on the underlying encoding

techniques and their formal foundations, which we present in what follows. To simplify this,

we apply it to Multi-Agent Path Finding (MAPF [14]), a simple yet highly relevant AI problem.

2. Background

We begin with some preliminaries from graph theory [15]. We consider graphs (𝑉,𝐸) where 𝑉
is a finite set of vertices and 𝐸 ⊆ 𝑉 ×𝑉 is a set of edges. A walk 𝜋 in a graph (𝑉,𝐸) is a sequence

(𝑣𝑖)
𝑛
𝑖=0 of vertices 𝑣𝑖 ∈ 𝑉 for 0 ≤ 𝑖 ≤ 𝑛 such that (𝑣𝑖, 𝑣𝑖+1) ∈ 𝐸 for all 0 ≤ 𝑖 < 𝑛. We use

𝜋(𝑖) = 𝑣𝑖 to refer to the vertex at index 0 ≤ 𝑖 ≤ 𝑛 of walk 𝜋 and |𝜋| = 𝑛 to refer to the length
of the walk. A walk 𝜋 in a graph (𝑉,𝐸) leads from 𝑢 ∈ 𝑉 to 𝑣 ∈ 𝑉 if 𝜋(0) = 𝑢 and 𝜋(𝑛) = 𝑣.

The vertex set of a walk 𝜋 is 𝑉 (𝜋) = {𝜋(𝑖) | 0 ≤ 𝑖 ≤ |𝜋|}; we write 𝑣 ∈ 𝜋 for 𝑣 ∈ 𝑉 (𝜋). The

index set of a walk 𝜋 in a graph (𝑉,𝐸) for a vertex 𝑣 ∈ 𝑉 is 𝜄𝜋(𝑣) = {𝑖 | 𝑣 = 𝜋(𝑖), 0 ≤ 𝑖 ≤ |𝜋|}.

A path is a walk in which all vertices (and therefore also all edges) are distinct. A cycle is a walk

in which only the first and last vertices are equal. Note that a cycle with just one vertex is also a

path. A stroll in a graph (𝑉,𝐸) is a walk in the reflexively closed graph (𝑉,𝐸∪{(𝑣, 𝑣) | 𝑣 ∈ 𝑉 }).
That is, a stroll is obtained from a walk in (𝑉,𝐸) by repeating some or none of its vertices

(to mimic waiting). A stroll 𝜋 is path-like, if 𝜄𝜋(𝑣) = [min 𝜄𝜋(𝑣),max 𝜄𝜋(𝑣)] for all 𝑣 ∈ 𝜋.

Informally, a stroll is path-like, if dropping all repeated vertices results in a path.

We use the above to define the MAPF problem. In what follows, we consider simple graphs

(𝑉,𝐸) where 𝐸 ⊆ 𝑉 × 𝑉 is irreflexive. A MAPF problem is a triple (𝑉,𝐸,𝐴) where (𝑉,𝐸)
is a simple graph and 𝐴 is a finite set of agents. Each agent 𝑎 ∈ 𝐴 has a start vertex 𝑠𝑎 ∈ 𝑉
and a goal vertex 𝑔𝑎 ∈ 𝑉 . We stipulate that all start and all goal vertices are disjoint. That

is, for all 𝑎, 𝑏 ∈ 𝐴, we require that 𝑎 ̸= 𝑏 implies 𝑠𝑎 ̸= 𝑠𝑏 and 𝑔𝑎 ̸= 𝑔𝑏. The start and goal

vertex of an agent may coincide, that is, we may have 𝑠𝑎 = 𝑔𝑎 for 𝑎 ∈ 𝐴. An agent can either

wait at its current vertex or move to a neighboring one. Hence, we use strolls to capture the

movement of agents. A plan of length 𝑛 for a MAPF problem (𝑉,𝐸,𝐴) is a family {𝜋𝑎}𝑎∈𝐴 of

strolls 𝜋𝑎 of length 𝑛 in (𝑉,𝐸) leading from 𝑠𝑎 to 𝑔𝑎 for all 𝑎 ∈ 𝐴. A plan for a MAPF problem

is path-based if all its strolls are path-like. We use 𝜄𝑎 as a shortcut for 𝜄𝜋𝑎 whenever it is clear

from context that agent 𝑎 is associated with stroll 𝜋𝑎. A plan {𝜋𝑎}𝑎∈𝐴 of length 𝑛 for a MAPF

problem (𝑉,𝐸,𝐴) is

1. vertex conflict-free if 𝜋𝑎(𝑖) ̸= 𝜋𝑏(𝑖) for all 𝑎, 𝑏 ∈ 𝐴 such that 𝑎 ̸= 𝑏 and 0 ≤ 𝑖 ≤ 𝑛,

2. swap conflict-free if 𝜋𝑎(𝑖 + 1) ̸= 𝜋𝑏(𝑖) or 𝜋𝑎(𝑖) ̸= 𝜋𝑏(𝑖 + 1) for all 𝑎, 𝑏 ∈ 𝐴 such that

𝑎 ̸= 𝑏 and 0 ≤ 𝑖 < 𝑛, and

3. follow conflict-free if 𝜋𝑎(𝑖+ 1) ̸= 𝜋𝑏(𝑖) for all 𝑎, 𝑏 ∈ 𝐴 such that 𝑎 ̸= 𝑏 and 0 ≤ 𝑖 < 𝑛.



A vertex conflict occurs if two agents occupy the same vertex at some point. A swap conflict

occurs if two agents traverse the same edge in opposite directions at some point. A follow

conflict occurs if an agent enters a vertex another agent just left. The absence of follow conflicts

implies the absence of swap conflicts.

Finally, we rely on a basic acquaintance with ASP, and refer the interested reader for details

to the literature [16]; the input language of clingo is described in the Potassco User Guide [17].

3. Routing

Consider the MAPF problem (𝑉,𝐸,𝐴) in Figure 1, delineating in blue and red the movements

Figure 1: Example MAPF problem together with stroll candidates

of agents 𝑎 and 𝑏 all of which traverse vertices 𝑣0,1 and 𝑣1,1 on the following strolls:

𝜋4
𝑎 = (𝑣0,2, 𝑣0,1, 𝑣1,1, 𝑣1,2, 𝑣1,3) 𝜋4

𝑏 = (𝑣1,0, 𝑣1,1, 𝑣0,1, 𝑣0,0, 𝑣0,0) (1)

𝜋5
𝑎 = (𝑣0,2, 𝑣0,1, 𝑣1,1, 𝑣1,2, 𝑣1,3, 𝑣1,3) 𝜋5

𝑏 = (𝑣1,0, 𝑣1,0, 𝑣1,0, 𝑣1,1, 𝑣0,1, 𝑣0,0) (2)

𝜋6
𝑎 = (𝑣0,2, 𝑣0,1, 𝑣1,1, 𝑣1,2, 𝑣1,3, 𝑣1,3, 𝑣1,3) 𝜋6

𝑏 = (𝑣1,0, 𝑣1,0, 𝑣1,0, 𝑣1,0, 𝑣1,1, 𝑣0,1, 𝑣0,0) (3)

Consider the following plans: {𝜋4
𝑎, 𝜋

4
𝑏} has length 4 and has a swap conflict, viz. 𝜋𝑎(1) = 𝜋𝑏(2)

and 𝜋𝑎(2) = 𝜋𝑏(1), plan {𝜋5
𝑎, 𝜋

5
𝑏} has length 5 and has no vertex- and swap-conflicts but a

follow conflict, viz. 𝜋𝑎(2) = 𝜋𝑏(3) (its moves are shown in Figure 2), and plan {𝜋6
𝑎, 𝜋

6
𝑏} has

length 6 and is conflict-free. (cf. Figure 3).

Figure 2: The plan {𝜋5
𝑎, 𝜋

5
𝑏} from (2) avoids vertex and swap conflicts but has a follow-conflict

3.1. Event orderings

Next, we characterize plans for MAPF problems in terms of orders on agent positions.

Figure 3: The plan {𝜋6
𝑎, 𝜋

6
𝑏} from (3) avoids vertex, swap, and follow conflicts

An event set ℰ for a MAPF problem (𝑉,𝐸,𝐴) is a set of events of form 𝑎@𝑣 where 𝑎 ∈ 𝐴 and

𝑣 ∈ 𝑉 . For illustration, consider the events corresponding to the positions of agents 𝑎 and 𝑏 in

Figures 2 and 3 (see also the plans in (2) to (3)):

ℰ𝑎 = {𝑎@𝑣0,2, 𝑎@𝑣0,1, 𝑎@𝑣1,1, 𝑎@𝑣1,2, 𝑎@𝑣1,3} (4)

ℰ𝑏 = {𝑏@𝑣1,0, 𝑏@𝑣1,1, 𝑏@𝑣0,1, 𝑏@𝑣0,0} (5)

Clearly, the combination ℰ𝑎 ∪ ℰ𝑏 is also an event set.

An ordered event set (ℰ ,≺) is a pair of events ℰ for a MAPF problem (𝑉,𝐸,𝐴) and a relation ≺
establishing a partial order among the events. We use ≺· to denote the cover of ≺. That is, ≺·
is the smallest relation such that ≺·* = ≺ where ·* is the transitive closure. The restriction of

an ordered event set (ℰ ,≺) for a MAPF problem (𝑉,𝐸,𝐴) to a single agent 𝑎 ∈ 𝐴 is denoted

(ℰ𝑎,≺𝑎) where ℰ𝑎 = {𝑎@𝑣 ∈ ℰ} and ≺𝑎 = ≺ ∩ (ℰ𝑎 × ℰ𝑎).
Taking a total order whose cover corresponds to the moves of agents 𝑎 and 𝑏 in Figures 2

and 3 results in totally ordered event sets (ℰ𝑎,≺𝑎) and (ℰ𝑏,≺𝑏) for both agents:

𝑎@𝑣0,2 ≺·𝑎 𝑎@𝑣0,1 ≺·𝑎 𝑎@𝑣1,1 ≺·𝑎 𝑎@𝑣1,2 ≺·𝑎 𝑎@𝑣1,3 (6)

𝑏@𝑣1,0 ≺·𝑏 𝑏@𝑣1,1 ≺·𝑏 𝑏@𝑣0,1 ≺·𝑏 𝑏@𝑣0,0 (7)

As above, the combination (ℰ𝑎 ∪ ℰ𝑏,≺𝑎 ∪ ≺𝑏) is also an ordered event set.

Definition 1. An ordered event set (ℰ ,≺) for a MAPF problem (𝑉,𝐸,𝐴) is path-based if

a) (ℰ𝑎,≺𝑎) is totally ordered with min≺𝑎 ℰ𝑎 = 𝑎@𝑠𝑎 and max≺𝑎 ℰ𝑎 = 𝑎@𝑔𝑎, and

b) (𝑢, 𝑣) ∈ 𝐸 for all 𝑎@𝑢, 𝑎@𝑢 ∈ ℰ with 𝑎@𝑢 ≺·𝑎 𝑎@𝑣.

For the MAPF problem (𝑉,𝐸,𝐴) depicted in Figure 1, the ordered event set (ℰ𝑎∪ℰ𝑏,≺𝑎∪≺𝑏)
is path-based since it contains the totally ordered event sets (ℰ𝑎,≺𝑎) and (ℰ𝑏,≺𝑏) with least

and greatest elements corresponding to the start and goal positions of agents 𝑎 and 𝑏 and their

covers agree with the edges of the graph (𝑉,𝐸).

Definition 2. An ordered event set (ℰ ,≺) for a MAPF problem (𝑉,𝐸,𝐴) is conflict-free if for
all events 𝑎@𝑢, 𝑏@𝑢 ∈ ℰ with 𝑎 ̸= 𝑏 either

a) there exists an event 𝑎@𝑣 ∈ ℰ such that 𝑎@𝑢 ≺·𝑎 𝑎@𝑣 and 𝑎@𝑣 ≺ 𝑏@𝑢, or



b) there exists an event 𝑏@𝑣 ∈ ℰ such that 𝑏@𝑢 ≺·𝑏 𝑏@𝑣 and 𝑏@𝑣 ≺ 𝑎@𝑢.

In fact, the ordered event set (ℰ𝑎∪ℰ𝑏,≺𝑎∪≺𝑏) comprises two conflicts because 𝑎@𝑣0,1, 𝑏@𝑣0,1 as

well as 𝑎@𝑣1,1, 𝑏@𝑣1,1 belong to ℰ𝑎∪ℰ𝑏 but are unrelated by ≺𝑎∪≺𝑏. The adjacency of 𝑣0,1 and

𝑣1,1 allows us to resolve this by adding 𝑎@𝑣1,2 ≺ 𝑏@𝑣1,1 (since this implies 𝑎@𝑣1,1 ≺ 𝑏@𝑣0,1).

That is, the ordered event set

(ℰ𝑎 ∪ ℰ𝑏, (≺𝑎 ∪ ≺𝑏 ∪ {𝑎@𝑣1,2 ≺ 𝑏@𝑣1,1})*) (8)

for the MAPF problem from Figure 1 is (path-based and) conflict-free.

The next definition establishes a general relation between path-based ordered event sets.

Definition 3. Two ordered event sets (ℰ ,≺1) and (ℰ ,≺2) for a MAPF program (𝑉,𝐸,𝐴) are
compatible if

a) 𝑎@𝑢 ≺1 𝑎@𝑣 iff 𝑎@𝑢 ≺2 𝑎@𝑣 for all 𝑎@𝑢, 𝑎@𝑣 ∈ ℰ , and

b) 𝑎@𝑢 ≺1 𝑏@𝑢 iff 𝑎@𝑢 ≺2 𝑏@𝑢 for all 𝑎@𝑢, 𝑏@𝑢 ∈ ℰ and 𝑎, 𝑏 ∈ 𝐴 with 𝑎 ̸= 𝑏.

In other words, two ordered event sets are compatible, if they resolve all conflicts among agents

in the same way.

We consider the ordered event set in (8) together with the following two:

(ℰ𝑎 ∪ ℰ𝑏, (≺𝑎 ∪ ≺𝑏 ∪ {𝑎@𝑣1,3 ≺ 𝑏@𝑣1,1})*) (9)

(ℰ𝑎 ∪ ℰ𝑏, (≺𝑎 ∪ ≺𝑏 ∪ {𝑏@𝑣0,0 ≺ 𝑎@𝑣0,1})*) (10)

All three ordered event sets are path-based and conflict-free. However, they differ regarding

compatibility. The ordered event sets in (8) and (9) are compatible; agent 𝑏 is waiting longer in

(9) before starting to move. The ordered event sets in (8) and (9) are both incompatible with the

one in (10); in the former two, agent 𝑎 starts moving before agent 𝑏 and in the latter, agent 𝑏
starts moving before agent 𝑎.

An ordered event set (ℰ ,≺1) is smaller than (ℰ ,≺2) if ≺1 ⊆ ≺2. This allows us to distinguish

minimally ordered event sets.

Proposition 1. For each conflict-free path-based ordered event set for a MAPF problem, there is a
unique minimally compatible conflict-free path-based ordered event set for the problem.

We call such an event set a minimal conflict-free path-based ordered event set.

The ordered event set in (8) is smaller than the one in (9). In fact, the one in (8) is a minimal

conflict-free path-based ordered event set.

Proposition 2. We have an onto relationship between plans and ordered event sets for MAPF
problems:

a) for each conflict-free path-based plan, there exists exactly one minimal conflict-free path-
based ordered event set, and



b) for each minimal conflict-free path-based ordered event set, there exists at least one conflict-
free path-based plan.

This onto relationship underlines the role of conflict-free path-based ordered event sets as an

abstraction of conflict-free path-based plans; the essence lies in the order of events, no matter

the continuance in a position. For example, the conflict-free path-based plan {𝜋6
𝑎, 𝜋

6
𝑏} from (3)

induces the minimal conflict-free path-based ordered event set in (8) and vice versa. The same

applies to extensions of {𝜋6
𝑎, 𝜋

6
𝑏} repeating vertices, as long as they preserve the order in (8).

Let us now detail the constructions underlying Proposition 2.a) and 2.b).

Given a conflict-free path-based plan {𝜋𝑎}𝑎∈𝐴 for a MAPF problem (𝑉,𝐸,𝐴), the corre-

sponding minimal conflict-free path-based ordered event set is the smallest ordered event set

({𝑎@𝜋𝑎(𝑖) | 𝑎 ∈ 𝐴, 0 ≤ 𝑖 ≤ |𝜋𝑎|},≺) satisfying the following conditions:

1. 𝑎@𝜋𝑎(𝑖) ≺ 𝑎@𝜋𝑎(𝑖+ 1) for all 0 < 𝑖 ≤ 𝑛 with 𝜋𝑎(𝑖) ̸= 𝜋𝑎(𝑖+ 1), and

2. 𝑎@𝜋𝑎(𝑖+ 1) ≺ 𝑏@𝜋𝑏(𝑗) for all 𝑎, 𝑏 ∈ 𝐴 and 0 ≤ 𝑖 < 𝑗 ≤ 𝑛 with 𝑎 ̸= 𝑏 and 𝜋𝑎(𝑖) =
𝜋𝑏(𝑗) ̸= 𝜋𝑎(𝑖+ 1).

Observe that the ordered event set in (8) satisfies the above conditions for plan {𝜋6
𝑎, 𝜋

6
𝑏}

given in (3). In fact, it is the smallest ordered event set satisfying them.

Given a conflict-free path-based ordered event set (ℰ ,≺) for a MAPF problem (𝑉,𝐸,𝐴), a

conflict-free path-based plan {𝜋𝑎}𝑎∈𝐴 is constructed as follows. First of all, we define a sequence

of mappings from events to time steps: for each 𝜖 ∈ ℰ and 𝑖 > 0, define

1. 𝛼0(𝜖) = 0

2. 𝛼𝑖(𝜖) = max{𝛼𝑖−1(𝜖)} ∪ {𝛼𝑖−1(𝜖
′) + 1 | 𝜖′ ∈ ℰ , 𝜖′ ≺· 𝜖}

Consider a fixed point 𝛼 = 𝛼𝑖 of this construction satisfying 𝛼𝑖 = 𝛼𝑖+1 for some 𝑖 ≥ 0.

Intuitively, 𝛼 gives the earliest arrival time of each agent at its traversed locations. With it, we

define for each agent 𝑎 ∈ 𝐴 its stroll 𝜋𝑎 = (𝑢𝑖)
𝑛
𝑖=0 as follows:

1. 𝑛 = max{𝛼(𝜖) | 𝜖 ∈ ℰ}

2. 𝑢𝑗 = 𝑣 for all 𝑎@𝑣 ≺· 𝑎@𝑤 and all 𝛼(𝑎@𝑣) ≤ 𝑗 < 𝛼(𝑎@𝑤)

3. 𝑢𝑗 = 𝑣 for 𝑎@𝑣 = max≺𝑎 ℰ𝑎 and all 𝛼(𝑎@𝑣) ≤ 𝑗 ≤ 𝑛

The resulting plan {𝜋𝑎}𝑎∈𝐴 is conflict-free and path-based. Note that there is no shorter plan

corresponding to the order since agents move as early as possible.

Consider the construction of arrival time mappings in Table 1 for the ordered event set in (8).

Each row corresponds to a mapping, where the first column is the running index 𝑖 and the

remaining ones give the arrival time of an agent at a vertex as indicated by the table header. We

obtain that 𝛼 = 𝛼6 is a fixed point. Observe that we can use this mapping to construct the plan

{𝜋6
𝑎, 𝜋

6
𝑏} given in (3). To verify, note that min 𝜄𝑐(𝑣) = 𝛼(𝑣@𝑐) for all 𝑐 ∈ {𝑎, 𝑏} and 𝑣 ∈ 𝜋6

𝑐 .



Table 1
Arrival time mapping construction for the ordered event set in (8)

𝛼𝑖 agent 𝑎 agent 𝑏

𝑖 𝑣0,2 𝑣0,1 𝑣1,1 𝑣1,2 𝑣1,3 𝑣1,0 𝑣1,1 𝑣0,1 𝑣0,0

0 0 0 0 0 0 0 0 0 0
1 0 1 1 1 1 0 1 1 1
2 0 1 2 2 2 0 2 2 2
3 0 1 2 3 3 0 3 3 3
4 0 1 2 3 4 0 4 4 4
5 0 1 2 3 4 0 4 5 5
6 0 1 2 3 4 0 4 5 6

3.2. Fact format

We represent a MAPF problem (𝑉,𝐸,𝐴) as a set of facts Γ(𝑉,𝐸,𝐴) consisting of an atom

vertex(𝑣) for each vertex 𝑣 ∈ 𝑉 , an atom edge(𝑢,𝑣) for each edge (𝑢, 𝑣) ∈ 𝐸, and atoms

agent(𝑎), start(𝑎,𝑠𝑎), and goal(𝑎,𝑔𝑎) for each agent 𝑎 ∈ 𝐴. We assume that a suitable

syntactic representation is chosen for values of italic variables; this representation is set in

typewriter in the source code.

We represent plans by predicates move/3 or move/4, depending on whether we use explicit

time points or not. In both cases, the first argument identifies an agent, and the second and

third an edge. A path or stroll (𝑣𝑖)
𝑛
𝑖=0 of an agent 𝑎 is then represented by atoms of form

move(𝑎,𝑣𝑖−1,𝑣𝑖) or move(𝑎,𝑣𝑖−1,𝑣𝑖,𝑖) with 𝑣𝑖−1 ̸= 𝑣𝑖 for 0 < 𝑖 ≤ 𝑛, respectively. We use

the variant with time points in the next section and drop them in Section 3.4.

3.3. Vanilla Encoding for MAPF

We first present a basic encoding for MAPF following the traditional approach in Answer Set

Planning [5], which relies on time steps.

The encoding in Listing 1 computes conflict-free plans {𝜋𝑎}𝑎∈𝐴 of length 𝑛 for MAPF

problems (𝑉,𝐸,𝐴). The parameter 𝑚 allows for preventing follow conflicts when set to fc.

We use a choice rule in Line 1 to generate move candidates. For each time point 1 ≤ 𝑡 ≤ 𝑛 and

each agent 𝑎 ∈ 𝐴, we choose at most one move(𝑎,𝑢,𝑣,𝑡) for an edge (𝑢, 𝑣) ∈ 𝐸. The selected

moves indicate agents 𝑎 exiting vertex 𝑢 at time point 𝑡−1 and arriving at vertex 𝑣 at time point 𝑡.
In Lines 3, 4, and 5, we generate candidates for strolls 𝜋𝑎 from the start positions 𝑠𝑎 of agents

𝑎 ∈ 𝐴 and moves selected in Line 1. Line 3 ensures that 𝜋𝑎(0) = 𝑠𝑎 encoded by at(𝑎,𝑠0,0).

In the following line, we derive agent positions at(𝑎,𝑣,𝑡) from move(𝑎,𝑢,𝑣,𝑡) establishing

𝜋𝑎(𝑡) = 𝑣. The last line in the block encodes that an agent stays at its position if it has not been

moved. That is, if we have 𝜋𝑎(𝑡− 1) = 𝑢 and the agent is not moved at time point 𝑡, we obtain

𝜋𝑎(𝑡) = 𝑢 captured by at(𝑎,𝑢,𝑡). Note the use of _ in the negative literal. This can be seen as

a shortcut for not move(A,U,T) together with the rule move(A,U,T) :- move(A,U,V,T)
projecting out variable V; the _ must not be interpreted as an anonymous variable here. At this

point, the vertices in the stroll candidates 𝜋𝑎 are not necessarily connected by edges nor do



Listing 1
Encoding to find bounded length plans for MAPF

1 { move(A,U,V,T): edge(U,V) } <= 1 :- agent(A), T=1..𝑛.

3 at(A,U,0) :- start(A,U).
4 at(A,V,T) :- move(A,_,V,T), T=1..𝑛.
5 at(A,U,T) :- at(A,U,T-1), not move(A,U,_,T), T=1..𝑛.

7 :- move(A,U,_,T), not at(A,U,T-1).
8 :- goal(A,U), not at(A,U,𝑛).

10 :- { at(A,U,T) } > 1, vertex(U), T=0..𝑛.
11 :- move(_,U,V,T), move(_,V,U,T).
12 :- at(A,U,T), at(B,U,T+1), A!=B, 𝑚=fc.

14 :- { at(A,U,T) } != 1, agent(A), T=1..𝑛.

they lead to goal vertices 𝑔𝑎. The next two integrity constraints in Lines 7 and 8 ensure that the

candidates 𝜋𝑎 are indeed strolls. Connectedness is ensured by the first integrity constraint. We

discard candidates whenever the agent is not at the source vertex of a move. For each true atom

move(𝑎,𝑢,𝑣,𝑡), we require 𝜋𝑎(𝑡−1) = 𝑢 by discarding solutions where not at(𝑎,𝑢,𝑡− 1)
is true. At this point, the stroll candidates are indeed strolls. The only missing piece is to require

that they lead to goal vertices. This is addressed by the second integrity constraint ensuring that

𝜋𝑎(𝑛) = 𝑔𝑎. For now, we have a plan {𝜋𝑎}𝑎∈𝐴 that is not necessarily conflict-free. In Lines 10,

11, and 12, we ensure that the plan is free of vertex, swap, and follow conflicts, respectively.

The first integrity constraint discards plans in which two agents occupy the same vertex at the

same time. The next integrity constraint ensures that there are no swap conflicts. Here, we

deviate slightly from the definition of swap conflicts. Observe that a swap conflict can only

occur if two agents travel between two vertices in opposing directions at the same time point.

Thus, we discard solutions with opposing moves. We treat follow conflicts in Line 12 following

the definition. Finally, we add a redundant check in Line 14 asserting that any agent must be at

exactly one position. This check is meant to improve solving performance.

Proposition 3 (Soundness). Given a MAPF problem (𝑉,𝐸,𝐴), let𝑃 be the union of Γ(𝑉,𝐸,𝐴)
and the program in Listing 1 for some 𝑛 ∈ N0 (and 𝑚 = fc).

If 𝑋 is an answer set of 𝑃 , then {(𝑣𝑖 | at(𝑎,𝑣𝑖,𝑖) ∈ 𝑋)𝑛𝑖=0}𝑎∈𝐴 is a vertex, swap, (and follow)
conflict-free plan for (𝑉,𝐸,𝐴).

Proposition 4 (Completeness). Given a MAPF problem (𝑉,𝐸,𝐴), let 𝑃 be the union of
Γ(𝑉,𝐸,𝐴) and the program in Listing 1 for some 𝑛 ∈ N0 (and 𝑚 = fc).

If {𝜋𝑎}𝑎∈𝐴 is a vertex, swap (and follow) conflict-free plan for (𝑉,𝐸,𝐴), then Γ(𝑉,𝐸,𝐴) ∪
{at(𝑎,𝜋𝑎(𝑖),𝑖) | 𝑎 ∈ 𝐴, 0 ≤ 𝑖 ≤ 𝑛}∪{move(𝑎,𝜋𝑎(𝑖− 1),𝜋𝑎(𝑖),𝑖) | 0 < 𝑖 ≤ 𝑛, 𝜋𝑎(𝑖−1) ̸=
𝜋𝑎(𝑖)} is an answer set of 𝑃 .



3.4. Ordering Encoding for MAPF

We now present a solution to MAPF reflecting event orderings; it consists of Listing 2, 3, and 4.

Listing 2
Encoding to find candidate paths for MAPF problems

1 { move(A,U,V): edge(U,V) } <= 1 :- agent(A), vertex(V).
2 { move(A,U,V): edge(U,V) } <= 1 :- agent(A), vertex(U).
3 :- move(A,U,_), not start(A,U), not move(A,_,U).
4 :- move(A,_,U), not goal(A,U), not move(A,U,_).

6 :- start(A,U), move(A,_,U).
7 :- goal(A,U), move(A,U,_).
8 :- start(A,U), not goal(A,U), not move(A,U,_).
9 :- goal(A,U), not start(A,U), not move(A,_,U).

The first encoding in Listing 2 selects atoms move(𝑎,𝑢,𝑣) for agents 𝑎 ∈ 𝐴 and

edges (𝑢, 𝑣) ∈ 𝐸 for a MAPF problem (𝑉,𝐸,𝐴). Given a stable model 𝑋 of Listing 2, the

selected moves for each agent 𝑎 form a subgraph (𝑉𝑎, 𝐸𝑎) of (𝑉,𝐸) such that

1. 𝐸𝑎 = {(𝑢, 𝑣) | move(𝑎,𝑢,𝑣) ∈ 𝑋},

2. 𝑉𝑎 = {𝑠𝑎, 𝑔𝑎} ∪ {𝑢, 𝑣 | (𝑢, 𝑣) ∈ 𝐸𝑎},

3. deg−(𝑣) = deg+(𝑢) = 1 for all move(𝑎,𝑢,𝑣) ∈ 𝑋 ,

4. deg−(𝑢) = 1 if 𝑢 ̸= 𝑠𝑎 and deg+(𝑣) = 1 if 𝑣 ̸= 𝑔𝑎 for all move(𝑎,𝑢,𝑣) ∈ 𝑋 ,

5. deg−(𝑠𝑎) = deg+(𝑔𝑎) = 0, and

6. deg−(𝑔𝑎) = deg+(𝑠𝑎) = 1 if 𝑠𝑎 ̸= 𝑔𝑎.

Any vertex on a path different from the start vertex must have an in degree (deg−) of one in

(𝑉𝑎, 𝐸𝑎); any vertex on a path different from the goal vertex must have an out degree (deg+) of

one in (𝑉𝑎, 𝐸𝑎); the start and goal vertices must be the start and end of the path. Hence, the

subgraph for an agent 𝑎 consists of exactly one path leading from 𝑠𝑎 to 𝑔𝑎 and zero or more

separate cycles with a length greater than or equal to two.

To see this, let us go over the rules and see how they affect the in and out degrees of vertices.

The rules in Lines 1 and 2 generate atoms move(𝑎,𝑢,𝑣) for agents 𝑎 ∈ 𝐴 such that (𝑢, 𝑣) ∈ 𝐸.

Furthermore, they ensure that deg−(𝑣) ≤ 1 and deg+(𝑢) ≤ 1 for all move(𝑎,𝑢,𝑣). We

also get deg−(𝑣) ≥ 1 and deg+(𝑢) ≥ 1 for all move(𝑎,𝑢,𝑣). Thus, at this point, we have

deg−(𝑣) = deg+(𝑢) = 1 for all move(𝑎,𝑢,𝑣). Lines 3 and 4 ensure that deg−(𝑢) = 1 if

𝑢 ̸= 𝑠𝑎 and deg+(𝑣) = 1 if 𝑣 ̸= 𝑔𝑎 for all move(𝑎,𝑢,𝑣). Note that this uses deg−(𝑢) ≤ 1 and

deg+(𝑣) ≤ 1 ensured in Lines 1 and 2. Lines 6 and 7 ensure that deg−(𝑠𝑎) = deg+(𝑔𝑎) = 0.

Lines 8 and 9 ensure that deg−(𝑔𝑎) = deg+(𝑠𝑎) = 1 if 𝑠𝑎 ̸= 𝑔𝑎. Note that this uses deg−(𝑔𝑎) ≤
1 and deg+(𝑠𝑎) ≤ 1 ensured in Lines 1 and 2.



The second encoding in Listing 3 selects atoms resolve(𝑎,𝑏,𝑢) in accord with Definition 2.

When such an atom is derived, agent 𝑎 has to depart from vertex 𝑢 before agent 𝑏 arrives at 𝑢.

The rule in Line 4 chooses which of two agents moving to the same vertex has to move first.

Listing 3
Encoding to find candidate event orders for MAPF problems

1 resolve(A,B,U) :- start(A,U), move(B,_,U), A!=B.
2 resolve(A,B,U) :- goal(B,U), move(A,_,U), A!=B.
3 { resolve(A,B,U);
4 resolve(B,A,U) } >= 1 :- move(A,_,U), move(B,_,U), A<B.

6 :- resolve(A,B,U), resolve(B,A,U).

The encoding assumes that 𝑠𝑎 ̸= 𝑠𝑏 and 𝑔𝑎 ̸= 𝑔𝑏 for all 𝑎, 𝑏 ∈ 𝐴 with 𝑎 ̸= 𝑏. Thus, a conflict

at a start and goal vertex can only arise if another agent moves to such a vertex. The rules in

Lines 1 and 2 select the right resolution order at these vertices: agents at their start vertex as

well as agents passing through the goal position of another agent have to move first. At this

point, there is at least one resolve atom for each case in Definition 2. The integrity constraint

in Line 6 ensures that there is exactly one. Note that there is exactly one true move(𝑎,𝑢,𝑣) for

each true resolve(𝑎,𝑏,𝑢). This is obvious for atoms derived by the rule in Line 4. Assume

that resolve(𝑎,𝑏,𝑢) is derived by the rule in Line 1. Then there are two cases. If 𝑠𝑎 = 𝑔𝑎, a

conflicting atom would be derived in Line 2. If 𝑠𝑎 ̸= 𝑔𝑎, there must be a move for agent 𝑎. The

same argument can be made for resolve atoms derived by the rule in Line 2.

Finally, the encoding in Listing 4 ensures that the move and resolve atoms from Listings 2

and 3 form a conflict-free path-based ordered event set. The edge directive in Line 1 specifies

Listing 4
Encoding to find conflict-free ordered event sets for MAPF

1 #edge ((A,U),(A,V)) : move(A,U,V).
2 #edge ((A,V),(B,U)) : resolve(A,B,U), move(A,U,V).

a graph containing edges given by the move atoms. We can interpret the tuples (𝑎, 𝑢) in the

edge right after the #edge keyword as events 𝑎@𝑢. Thus for each true move(𝑎,𝑢,𝑣) an

edge (𝑎@𝑢, 𝑎@𝑣) is added to the graph. The solver discards all solutions where this graph

contains a cycle. Remember that the encoding in Listing 2 admits graphs with paths and cycles

for agents. Thus, at this point, we have ensured that the moves form a path-based ordered event

set. The edge directive in Line 2 further extends the above graph. In accord with Definition 2,

we add edge (𝑎@𝑣, 𝑏@𝑢) for each atom resolve(𝑎,𝑏,𝑢) where 𝑎@𝑢 ≺·𝑎 𝑎@𝑣 is captured by

move(𝑎,𝑢,𝑣). We have seen above that there is exactly one such move. Thus, if the resulting

graph is acyclic, then there is a corresponding conflict-free path-based ordered event set.

Next, we define how to obtain orders from answers sets of the above programs.

Definition 4. Let (𝑉,𝐸,𝐴) be a MAPF problem and 𝑋 be an answer set of Γ(𝑉,𝐸,𝐴) and the
programs in Listings 2, 3, and 4.

The ordered event set (ℰ ,≺) corresponding to 𝑋 is the smallest ordered event set such that



a) ℰ = {𝑎@𝑢, 𝑎@𝑣 | move(𝑎,𝑢,𝑣) ∈ 𝑋} ∪ {𝑏@𝑢 | resolve(𝑎,𝑏,𝑢) ∈ 𝑋}

b) 𝑎@𝑢 ≺ 𝑎@𝑣 for all move(𝑎,𝑢,𝑣) ∈ 𝑋 , and

c) 𝑎@𝑣 ≺ 𝑏@𝑢 for all resolve(𝑎,𝑏,𝑢), move(𝑎,𝑢,𝑣) ∈ 𝑋 .

Proposition 5 (Soundness). Let (𝑉,𝐸,𝐴) be a MAPF problem and 𝑃 be the union of
Γ(𝑉,𝐸,𝐴) and the programs in Listings 2, 3, and 4.

If 𝑋 is an answer set of 𝑃 , then the ordered event set (ℰ ,≺) corresponding to 𝑋 is a minimal
conflict-free path-based ordered event set for (𝑉,𝐸,𝐴).

Together with Proposition 2.b), this implies the existence of a corresponding path-based plan,

which can be constructed as described at the end of Section 3.1.

Definition 5. Let (𝑉,𝐸,𝐴) be a MAPF problem and (ℰ ,≺) be a minimal conflict-free path-based
ordered event set for (𝑉,𝐸,𝐴).

We define the set 𝑋 of atoms corresponding to (ℰ ,≺) as the smallest set such that

a) Γ(𝑉,𝐸,𝐴) ⊆ 𝑋 ,

b) move(𝑎,𝑢,𝑣) ∈ 𝑋 for all 𝑎@𝑢 ≺·𝑎 𝑎@𝑣, and

c) resolve(𝑎,𝑏,𝑢) ∈ 𝑋 for all 𝑎@𝑣 ≺ 𝑏@𝑢 and 𝑎@𝑢 ≺·𝑎 𝑎@𝑣 with 𝑎 ̸= 𝑏.

Proposition 6 (Completeness). Let (𝑉,𝐸,𝐴) be a MAPF problem and 𝑃 be the union of
Γ(𝑉,𝐸,𝐴) and the programs in Listings 2, 3, and 4.

If (ℰ ,≺) is a minimal conflict-free path-based ordered event set for (𝑉,𝐸,𝐴), then the set of
atoms corresponding to (ℰ ,≺) is an answer set of 𝑃 .

This and Proposition 2.a) implies that each path-based plan of a MAPF problem has a corre-

sponding stable model of the MAPF encodings.

4. Experiments

To evaluate our encoding variants, we built an ASP-based benchmark generator for different

types of grid-based MAPF problems,
1

viz. random and room configurations. The instances are

created by choosing subgraphs of a square grid graph (𝑉,𝐸) with 𝑉 = {𝑣𝑖𝑗 | 1 ≤ 𝑖, 𝑗 ≤ 𝑛}
and 𝐸 = {(𝑣𝑖𝑗 , 𝑣𝑘𝑙) ∈ 𝑉 × 𝑉 | |𝑖− 𝑘|+ |𝑗 − 𝑙| = 1} for some 𝑛 > 0. The generated graphs

are undirected and connected. Agents’ start and goal vertices are picked randomly. For room

instances, the grid is evenly divided into rooms separated by walls of width one; neighboring

rooms are connected by choosing up to one random vertex as a door. For random ones, a certain

percentage of the vertices of the square grid graph is chosen.

Instances are built via the following parameters: 10 ≤ 𝑛 ≤ 40 for the size of the underlying

square grid graph, between 5 and 30 agents, square rooms of size 1 to 5, and vertices are

chosen with a 50% probability for the random instances. In total, we consider 105 instances

1https://github.com/krr-up/mapf-instance-generator

https://github.com/krr-up/mapf-instance-generator

Figure 4: Evaluation of random and room instances with path-based plans

(56 random, 38 room) with at least one conflict-free path-based plan. We ran all instances on

a compute cluster with Intel Xeon E5-2650v4@2.9GHz CPUs with 64GB of memory running

Debian Linux 10.
2

We used a timeout of 3h and limited the memory to 16GB per instance.

We consider the vanilla encoding (vanilla) in Listing 1 and the event ordering encoding (order
AC) described in Section 3.4. Furthermore, we use a variant of the latter replacing the acyclicity

constraints in Listing 4 with difference constraints (order DC) and clingo[dl] as solver:

1 &diff{(A,U)+1}<=(A,V) :- move(A,U,V).
2 &diff{(A,V)+1}<=(B,U) :- resolve(A,B,U), move(A,U,V).

Finally, we use a plain ASP encoding (order ASP) to check if the generated order is acyclic.

Without going into details, we mention that this check results in groundings of cubic size in the

number of possible events.

We try to find both conflict-free (vsfc-free) and vertex and swap conflict-free plans (vsc-free).

In the second setting, this means setting the mode 𝑚 to fc for the vanilla encoding. For the

DC encoding, this means changing the difference constraint in Line 1 above to use +0 instead

of +1 and an additional integrity constraint to discard swap conflicts. Note that the integrity

constraint does not increase the asymptotic size complexity of the grounding. For the ASP and

AC encodings, the second setting involves a more complicated construction, which we do not

detail here. Note that even for the AC encodings, this involves a lot more overhead comparable

to that of the order ASP setting.

We formulate the following hypotheses for evaluating the generated benchmark on the

configurations described above. The vanilla encoding should perform well for small plan

lengths (H1). The order ASP configuration is not applicable in practice because of the too large

grounding (H2). The order AC configuration works well for path-based plans in the vsfc-free

setting even with large plan lengths (H3) but does not work well in the vsc-free setting (H4). The

order DC configuration works well for path-based plans with large plan lengths independent of

the conflict handling (H5).

The plots in Figure 4 present cactus plots showing the time (x-axis) to solve an instance (y-

2https://www.cs.uni-potsdam.de/bs/research/labs.html#hardware

https://www.cs.uni-potsdam.de/bs/research/labs.html#hardware


axis); run times are sorted in ascending order for each curve matching one of the configurations

discussed above. Note that we fix the plan length for the vanilla encoding in the vsc-free setting

to the optimal length such that an instance is satisfiable. Since we did not have this information

available for vsfc-free (and it is hard to compute), we added 20% to the plan length.

We begin with evaluating the vsc-free setting. In line with H1, we observe that the vanilla

encoding performs well noting that agents can move to their goal vertices rather directly.

Agreeing with H5, order DC performs well too. On room instances it is better than vanilla,

which we attribute to the longer plan length necessary to avoid obstacles. As stipulated in H2

and H4, the order ASP and order AC configurations do not work at all.

In the vsfc-free setting, we evaluate the random and room instances separately. We begin with

the random instances. Note that order DC performs almost the same as in the vsc-free setting.

However, the vanilla configuration becomes much slower if plans have to be follow conflict-free.

While this is partly associated with larger plan lengths, we cannot fully explain this behavior.

Finally, we confirm H3, noting that order AC performs well here. However, despite the less

expressive AC constraints, it performs does not perform as well as order DC. This is due to

clingo[dl] implementing stronger propagation enabled using option --propagate=full. The

room instances are much harder in the vsfc-free setting. Not just vanilla but also the event

order based configurations are slowed down. We attribute this to short plan length due to H1

and large overhead due to many reachable vertices for the event order settings.

5. Discussion

Although our focus has been on routing, the addition of scheduling boils down to replacing

the acyclicity constraints in Listing 4 by difference constraints taking durations into account.

Assuming that the edge predicate has a third duration argument, we use the following rules:

1 &diff{(A,U)+D}<=(A,V) :- move(A,U,V), edge(U,V,D).
2 &diff{(A,V)+D}<=(B,U) :- resolve(A,B,U), move(A,U,V), edge(U,V,D).

Without entering details, the above should suffice to indicate that our approach generalizes to

combined routing and scheduling in a seamless way (as detailed in the upcoming full version of

our paper). As mentioned, we have already successfully applied this technique in industrial

applications involving routing and scheduling [11, 12, 13]. This paper aims at providing an

introduction to the underlying encoding techniques and their formal foundations.

Our approach builds on the characterization of plans in terms of partially ordered event sets.

This is similar to partial order planning, where a partial order is maintained among actions [10].

In MAPF, a related idea was implemented using activity constraints from constraint-based

scheduling [18]. Also, simple temporal networks were used to add schedules to pre-computed

plans [19]. Similarly, in robotics, action dependency graphs were used to avoid online collisions

by adding temporal dependencies to existing plans [20]. Finally, it is worth mentioning that it

was recently shown that whenever each agent has a predefined path, the problem of deciding if

there is a MAPF solution (without any bounds on any cost function) is NP-Hard [21].

